version 18.0               // version control
set processors 8           // to ensure replicability across different numbers of cores
clear all                  // clear existing data
macro drop _all            // and macros, clean slate
set seed 20220909          // set seed
/* 
Notes: 
1) Must put "set sortseed #" right on top of EACH "telasso" (rather than at the top of the do file) to ensure the same number of selected controls will be reported.
2) For detailed explanations, see Stata's technical support's email chain on 10/10/2023. 
*/

local pgm  "dst-Table_3_Panel_C_sleep_deprivation_telasso_main_results" // file name
local who  "Muzhe Yang"                                                 // author
local dte  "2025-01-20"                                                 // created date
local dte2 "`c(current_date)'"                                          // last run date
local tag  "`pgm'.do, created by `who' on `dte', last run on `dte2'"

capture log close
log using "code\analysis\tables\\`pgm'.txt", text replace 
display "`tag'"

**# data prep ------------------------------------------------------------------------------------

use "data_clean\dst-data04_for_estimation_within_250_miles", clear
summ dist_to_border
tab wave, missing
preserve 
keep wave TractFIPS sleep_depr
reshape wide sleep_depr, i(TractFIPS) j(wave)
compare sleep_depr1 sleep_depr2  // two waves of data are identical, all from 2018
restore
keep if wave == 2
codebook TractFIPS

*## (1) create the 9 cells, following page 215 of the following paper:
*## Giuntella, O. and F. Mazzonna (2019). "Sunset Time and the Economic Effects of Social Jetlag: Evidence from US Time Zone Borders." Journal of Health Economics 65: 210-226.

assert !missing(centroid_lat)
assert !missing(region)
gen     cell = 1 if centroid_lat > 40                        & region == "eastern and central"
replace cell = 2 if (34 < centroid_lat & centroid_lat <= 40) & region == "eastern and central"
replace cell = 3 if centroid_lat <= 34                       & region == "eastern and central"
replace cell = 4 if centroid_lat > 40                        & region == "central and mountain"
replace cell = 5 if (34 < centroid_lat & centroid_lat <= 40) & region == "central and mountain"
replace cell = 6 if centroid_lat <= 34                       & region == "central and mountain"
replace cell = 7 if centroid_lat > 40                        & region == "mountain and pacific"
replace cell = 8 if (34 < centroid_lat & centroid_lat <= 40) & region == "mountain and pacific"
replace cell = 9 if centroid_lat <= 34                       & region == "mountain and pacific"
tab cell time_zone, missing
summ centroid_lat if cell == 1 | cell == 4 | cell == 7
summ centroid_lat if cell == 2 | cell == 5 | cell == 8
summ centroid_lat if cell == 3 | cell == 6 | cell == 9

*## (2) control variables

global x                          "pop white_pct black_pct hispanic_pct pop_under_18yr_pct pop_65yr_over_pct age_median educ_hs_pct educ_coll_pct married_pct hh_size hh_income_median home_value_median no_health_ins_pct unemploy_pct"
global x_with_density "pop pop_density white_pct black_pct hispanic_pct pop_under_18yr_pct pop_65yr_over_pct age_median educ_hs_pct educ_coll_pct married_pct hh_size hh_income_median home_value_median no_health_ins_pct unemploy_pct"
global cvars                                c.($x dist_to_border centroid_lat daylight_dur_max)##c.($x dist_to_border centroid_lat daylight_dur_max)
global cvars_with_density      c.($x_with_density dist_to_border centroid_lat daylight_dur_max)##c.($x_with_density dist_to_border centroid_lat daylight_dur_max)
global controls                           $cvars i.cell i.cell#c.($x dist_to_border centroid_lat daylight_dur_max) 
global controls_with_density $cvars_with_density i.cell i.cell#c.($x_with_density dist_to_border centroid_lat daylight_dur_max)

des  $x dist_to_border centroid_lat daylight_dur_max
summ $x dist_to_border centroid_lat daylight_dur_max

**# estimation prep -------------------------------------------------

encode CountyFIPS, gen(county_fips)
egen nomiss = rowmiss($x dist_to_border centroid_lat daylight_dur_max)
assert !missing(StateAbbr)
local radius = 50
gen sample_selection = (nomiss == 0 & StateAbbr != "AZ" & dist_to_border <= `radius')
global xfolds_resample "xfolds(10) resample(3) nolog"

**# table ---------------------------------------------------------------------------------

set sortseed 12102023
telasso (sleep_depr $controls) (treat $controls) if sample_selection == 1, vce(cluster county_fips) $xfolds_resample rseed(10101)
estimates store m1
summ sleep_depr if e(sample) == 1
scalar y_mean = r(mean)
scalar k_controls = e(k_controls)
scalar k_controls_sel = e(k_controls_sel)
scalar popden = "No"
quietly etable, replace ///
        column(index) ///
        keep(r1vs0.treat) ///
        cstat(_r_b,  nformat(%9.3f)) ///
        cstat(_r_se, nformat(%9.3f)) ///
        mstat(dep_mean = y_mean, nformat(%9.3f) label("Mean of the dependent variable")) ///
        mstat(effect_relative_magnitude = 100*_r_b[r1vs0.treat]/y_mean, nformat(%9.1f) label("(Estimate/mean) × 100%")) ///
        mstat(N, nformat(%9.0fc) label("Number of observations")) ///
        mstat(k_controls, nformat(%9.0fc) label("Number of potential predictor variables")) ///
        mstat(k_controls_sel, nformat(%9.0fc) label("Number of selected predictor variables")) ///
        mstat(case_popden = popden, label("Population density included in the set of potential predictor variables")) ///
        stars(0.10 "*" 0.05 "**" 0.01 "***", attach(_r_b) decreasing pvname("p-value") nformat(%9.2f)) showstars showstarsnote ///
        novarlabel nocenter 

set sortseed 12102023
telasso (sleep_depr $controls_with_density) (treat $controls_with_density) if sample_selection == 1, vce(cluster county_fips) $xfolds_resample rseed(10101)
estimates store m2
summ sleep_depr if e(sample) == 1
scalar y_mean = r(mean)
scalar k_controls = e(k_controls)
scalar k_controls_sel = e(k_controls_sel)
scalar popden = "Yes"
quietly etable, append

set sortseed 12102023
telasso (sleep_depr $controls) (treat $controls) if sample_selection == 1 & region == "eastern and central", vce(cluster county_fips) $xfolds_resample rseed(10101)
estimates store m3
summ sleep_depr if e(sample) == 1
scalar y_mean = r(mean)
scalar k_controls = e(k_controls)
scalar k_controls_sel = e(k_controls_sel)
scalar popden = "No"
quietly etable, append

set sortseed 12102023
telasso (sleep_depr $controls_with_density) (treat $controls_with_density) if sample_selection == 1 & region == "eastern and central", vce(cluster county_fips) $xfolds_resample rseed(10101)
estimates store m4
summ sleep_depr if e(sample) == 1
scalar y_mean = r(mean)
scalar k_controls = e(k_controls)
scalar k_controls_sel = e(k_controls_sel)
scalar popden = "Yes"
quietly etable, append

collect title "Table: The Impact of Circadian Misalignment Estimated by the Double/Debiased ML" 

collect layout 
collect addtags top_row[1]
collect recode etable_estimates 1 = column1 2 = column1 ///
                                3 = column2 4 = column2 
collect layout (colname[r1vs0.treat]#result[_r_b _r_se] result[dep_mean effect_relative_magnitude N k_controls k_controls_sel case_popden]) ///
               (top_row[1]#etable_estimates#cmdset#stars)

collect label levels top_row ///
        1 "Treat = 1 for census tracts located east of the time zone border; Treat = 0 for census tracts located west of the time zone border", modify
collect label levels etable_estimates ///
        column1 "All time zones in the Contiguous United States" ///
        column2 "Eastern Time Zone and Central Time Zone", modify 
collect label levels cmdset ///
        1 "(1)" 2 "(2)" 3 "(3)" 4 "(4)", modify
collect label levels colname r1vs0.treat "Treat (1/0)", modify 

collect style header top_row, level(label)
collect style header etable_estimates, level(label)
collect style header cmdset, level(label)
collect style header colname, level(label)
collect style cell result[_r_b dep_mean N k_controls k_controls_sel], sformat(%s)
collect style cell result[effect_relative_magnitude], sformat("%s%%")
collect style cell border_block[corner], border(top, pattern(double))
collect style cell border_block[column-header], border(top, pattern(double))
collect style cell border_block[row-header], border(bottom, pattern(double))
collect style cell border_block[item], border(bottom, pattern(double))

collect export "code\analysis\tables\\`pgm'.xlsx", replace

log close
exit